Introduzione a Maple

1 Calcoli numerici e algebrici

I comandi possono terminare con ;

```
> 3*2;
```

$$6$$

oppure con :

```
> 3+2:
```

in questo caso l'output non viene mostrato (si vedra' che questo e' comodo....) ma viene creato.

per vederlo, basta richiamarlo tramite "%" (che significa l'ultimo risultato calcolato)

```
> %;
```

$$5$$

Maple si ricorda tutti i risultati delle operazioni per tutta la sessione (%% e' il penultimo risultato calcolato, %%% il terzultimo... deve esistere una sintassi migliore per richiamare i risultati di molte operazioni prima, ma non lo conosco e non sarebbe molto utile)

```
> %%%; %%;
```

$$6$$
$$5$$

Se vogliamo far "dimenticare" tutto quanto fatto in precedenza (spesso e' utile), c'e' il comando "restart"

```
> restart; %%;
```

Operazioni

L'elevamento a potenza si fa con l'operatore ˆ

```
> 2^3;
```

$$8$$

Le operazioni si combinano, usando le parentesi, con le normali regole algebriche

```
> (4+3/7)^2/(1-9/4);
```

$$\frac{-3844}{245}$$

I calcoli vengono svolti esattamente con numeri interi e razionali.

Se invece vogliamo vederlo in forma decimale si usa il comando "evalf" (=eval float)

```
> evalf(%);
```

$$-15.68979592$$

La radice quadrata si calcola usando "sqrt" (quasi tutti le funzioni hanno un nome simile all'espressione inglese; se uno non si ricorda bene il nome, ma sa che in inglese si dice "square root" puo' scegliere Menu Help Topic Search e cominciare a scrivere le prime lettere per vedere i comandi esistenti e leggere cosa fanno)

```
> sqrt(2);
```

e la radice viene usata in modo simbolico esatto.

$$\sqrt{2}$$

Infatti

```
> %^2;
```

$$2$$

Se vogliamo saperne un valore numerico, usiamo "evalf" (%% e' il penultimo risultato calcolato)

```
> evalf(%%);
```

$$1.414213562$$

```
> %^2;
```

Ma a questo punto l'aritmetica non e' piu' esatta.

$$1.999999999$$

"evalf" puo' avere un secondo argomento che e' il numero di cifre decimali usate

```
> evalf(sqrt(2),50);
```

$$1.4142135623730950488016887242096980785696718753770$$

In Maple possiamo anche operare con simboli

```
> (x+y)^3;
```

$$(x+y)^3$$

e usare le regole dell'algebra:

```
> simplify(%);
```

In questo caso Maple ritiene che si tratti gia' della forma piu' semplice.

$$(x+y)^3$$

Allora possiamo espandere l'espressione

```
> expand(%);
```

$$x^3 + 3\,x^2\,y + 3\,x\,y^2 + y^3$$

Ci sono altri comandi per la manipolazione di espressioni simboliche, ma non e' semplice usarli ottenendone il risultato desiderato. Nel corso non affronteremo questo argomento in dettaglio.

2 Variabili

Come in tutti i linguaggi di programmazione, e' utile avere variabili a cui assegnare valori.

:= (da scrivere sempre attaccati) e' l'istruzione di assegnazione

```
> a := 2;
```

A questo punto ad "a" e' assegnato il valore 2

$$a := 2$$

```
> 3*a;
```

$$6$$

```
> sqrt(a);
```

$$\sqrt{2}$$

"=" e' il simbolo che si usa nelle equazioni, non nelle assegnazioni

```
> b = 2;
```

Questa e' un'equazione

$$b = 2$$

```
> b;
```

Infatti b e' ancora un nome

$$b$$

```
> a;
```

mentre "a" continua a valere 2

$$2$$

Per far tornare "a" semplicemente un nome possiamo fare cosi':

```
> a := 'a';
```

$$a := a$$

```
> a;
```

$$a$$

E' un comando molto utile (alterantivo all'azzeramento totale con "restart") perche' molti errori che si ottengono usando Maple derivano dal fatto che Maple si ricorda assegnazioni che abbiamo gia' fatto delle variabili che voliamo usare adesso...

Una variabile puo' essere di tipo insieme

```
> a:={7,31,giovanni,2,6};
```

$$a := \{2, 6, 7, 31, \mathit{giovanni}\}$$

Negli insiemi gli elementi ripetuti non vengono considerati

```
> a:={7,31,6,31,giovanni,giuseppe,giovanni};
```

$$a := \{6, 7, 31, \mathit{giovanni}, \mathit{giuseppe}\}$$

Questo e' il secondo elemento dell'insieme (negli insiemi non vi e' ordinamento, per cui Maple ordina gli elementi come gli pare...). Per indicare gli elementi di un insieme, una lista, un vettore... si usa [ ] come in C

```
> a[2];
```

$$7$$

Questa e' una lista (gli elementi vengono ripetuti)

```
> a:=[7,31,6,31,giovanni,giuseppe,giovanni];
```

$$a := [7, 31, 6, 31, \mathit{giovanni}, \mathit{giuseppe}, \mathit{giovanni}]$$

e il secondo elemento e' quello assegnato da noi

```
> a[2];
```

$$31$$

3 Funzioni e grafici

Possiamo creare una variabile che contiene un simbolo:

```
> f := exp(-x^2/10);
```

$$f := e^{(-1/10\,x^2)}$$

Se adesso diamo un valore al simbolo x

```
> x := 1;
```

$$x := 1$$

in f adesso x vale 1

```
> f;
```

$$e^{(\frac{-1}{10})}$$

ancora:

```
> x:= 0; f;
```

$$x := 0$$
$$1$$

adesso che x e' definita, se usiamo x in un'altra espressione g

```
> g := cos(2*x);
```

$$g := 1$$

g e' ormai definita cosi' e non varia piu', variando x

```
> x := 1; g;
```

$$x := 1$$
$$1$$

```
> x := 'x':  g:= 'g':
```

Il precedente non e' quindi il modo giusto per definire una funzione. Maple prevede proprio il concetto di funzione, analogo a quello usato in matematica:

x -> espressione e' il modo piu' semplice per definire una funzione ($e\hat{}x$ si scrive *exp(x)*)

```
> f := x -> exp(-x^2/10);
```

esiste anche una sintassi piu' informatica usando l'operatore "proc", ma per ora non ci servira"

$$f := x \to e^{(-1/10\,x^2)}$$

Controlliamo di aver fatto giusto:

```
> f(1);
```

$$e^{(\frac{-1}{10})}$$

Facciamo il grafico di f, usando il comando "plot" che e' molto intuitivo

```
> plot(f);
```



Questa sintassi (plot di un'espressione, specificando la variabile) e la precedente (plot di una funzione) sono equivalenti (l'importante e' non mescolare le due sintasssi):

```
> plot(f(x),x);
```



Se non viene detto niente, Maple sceglie [-10,10] come intervallo sull'asse delle ascisse. Se si vuole cambiare, basta assegnarlo esplicitamente

```
> plot(f,0..5);
```



oppure

```
> plot(f(x),x=0..5);
```



Come fa un grafico Maple? calcola la funzione (o l'espressione) in un certo numero di punti (modificabile tramite l'opzione "numpoints") e li unisce.

Se uno desidera, e' facile ingannare Maple ed ottenere grafici sbagliati:

```
> plot(f,-1e6..1e6);
```

Non si vede niente, perche' la funzione vale praticamente 0 in tutti i punti in cui e' calcolata...



Altri esempi:

```
> plot (cos(x^2+1)*exp(-0.3*x), x = 0..10);
```



```
>  plot (cos(x^2+1)*exp(-0.3*x), x = 0..10,y=-1..1);
```



Se si vogliono sovrapporre piu' grafici, si puo' fare una lista di espressioni di cui fare il plot (*Pi* e' la costante nota)

```
>  plot([cos(x),sin(x)],x=-Pi..Pi);
```

I colori sono scelti automaticamente da Maple, ma volendo si possono cambiare (cosi' come molte altre caratteristiche dei grafici)



Questo potrebbe sembrare il modo di definire una funzione:

```
>   g(x) := cos(3*x) - 2*x^2;
```

$$\mathrm{g}(x) := \cos(3\,x) - 2\,x^2$$

ma non lo e'.

```
>   g(1);
```

g(1) e' rimasto un nome...

$$\mathrm{g}(1)$$

...anche se g(x) e' quello che abbiamo definito

```
>   g(x);
```

ma non abbiamo detto che chiunque sia x quello dovra' essere il valore di g in x

$$\cos(3\,x) - 2\,x^2$$

Questo e' il modo corretto di definire "g"

```
>   g := x -> cos(3*x) - 2*x^2;
```

$$g := x \rightarrow \cos(3\,x) - 2\,x^2$$

4 Soluzione di equazioni

Ci sono due comandi per risolvere equazioni: "solve" e "fsolve".

"solve" usa metodi simbolici di risoluzione (come la formula per le soluzioni di equazioni di 2o grado); "fsolve" invece usa metodi numerici, ossia un algoritmo che, partendo da un punto iniziale, trova approssimazioni successive per una (o piu') soluzione usando calcoli con numeri.

iniziamo con "restart" che fa dimenticare a Maple tutto quanto svolto prima. E' utile per evitare che assegnazioni precedenti influiscano sui risultati

```
>   restart;
```

Cominciamo con "solve" e assegnamo un'equazione di secondo grado [la sintassi e' solve(equazione, variabile)]

```
>   solve(3*x^2-7*x+2 = 0, x);
```

$$2,\ \frac{1}{3}$$

Passiamo a un'equazione generale di 2o grado

```
>   solve(a*x^2+b*x+c=0,x);
```

$$\frac{1}{2}\,\frac{-b + \sqrt{b^2 - 4\,a\,c}}{a},\ \frac{1}{2}\,\frac{-b - \sqrt{b^2 - 4\,a\,c}}{a}$$

Maple conosce la formula risolutiva...

Per Maple possiamo far risolvere rispetto a x, oppure rispetto ad a

```
> solve(a*x^2+b*x+c=0,a);
```

$$-\frac{b\,x+c}{x^2}$$

Passiamo a u'equazione di terzo grado

```
> solve (x^3+ p*x+q=0,x);
```

$$\frac{1}{6}\,\%1^{(1/3)} - \frac{2\,p}{\%1^{(1/3)}},\ -\frac{1}{12}\,\%1^{(1/3)} + \frac{p}{\%1^{(1/3)}} + \frac{1}{2}\,I\,\sqrt{3}\,(\frac{1}{6}\,\%1^{(1/3)} + \frac{2\,p}{\%1^{(1/3)}}),$$
$$-\frac{1}{12}\,\%1^{(1/3)} + \frac{p}{\%1^{(1/3)}} - \frac{1}{2}\,I\,\sqrt{3}\,(\frac{1}{6}\,\%1^{(1/3)} + \frac{2\,p}{\%1^{(1/3)}})$$
$$\%1 := -108\,q + 12\,\sqrt{12\,p^3 + 81\,q^2}$$

Esiste una formula risolutiva un po' complicata. Notare che I e' l'unita' immaginaria (in genere scritta i) cioe' tale che
$I^2 = -1$

Proviamo a passare a un'equazione specifica:

```
> solve(x^3-7*x+4,x);
```

$$\frac{1}{3}\,\%1 + \frac{7}{(-54+3\,I\,\sqrt{705})^{(1/3)}},$$
$$-\frac{1}{6}\,\%1 - \frac{7}{2}\,\frac{1}{(-54+3\,I\,\sqrt{705})^{(1/3)}} + \frac{1}{2}\,I\,\sqrt{3}\,(\frac{1}{3}\,\%1 - \frac{7}{(-54+3\,I\,\sqrt{705})^{(1/3)}}),$$
$$-\frac{1}{6}\,\%1 - \frac{7}{2}\,\frac{1}{(-54+3\,I\,\sqrt{705})^{(1/3)}} - \frac{1}{2}\,I\,\sqrt{3}\,(\frac{1}{3}\,\%1 - \frac{7}{(-54+3\,I\,\sqrt{705})^{(1/3)}})$$
$$\%1 := (-54+3\,I\,\sqrt{705})^{(1/3)}$$

Proviamo a farci semplificare...

```
> simplify(%);

Error, (in simplify) invalid simplification command
```

Allora vediamo il valore numerico

```
> evalf(%);
```

$$2.292401586 + 0.\,I,\ -2.895106517 + .1\,10^{-9}\,I,\ .602704931 + .1\,10^{-9}\,I$$

Si vede che una radice e' reale (il coefficiente di I e' 0) [del resto, e' chiaro graficamente che ogni equazione di terzo grado ha almeno una radice reale]. Le altre due radici hanno un coefficiente di I estremamente piccolo dell'ordine di grandezza della precisione di "evalf": si puo' sospettare che il valore "vero" della parte immaginaria sia 0.

Allora valutiamo le soluzioni con 50 cifre decimali

```
> evalf(%%,50);
```

$$2.2924015852246210323037033576640610116122205124393 + 0.\,I,$$
$$-2.8951065159275306769348628012682378820513101595554-$$
$$.9660254037844386467637231707529361834714026269052\,10^{-49}\,I,$$
$$.6027049307029096446311594436041768704390896471160+$$
$$.7660254037844386467637231707529361834714026269052\,10^{-49}\,I$$

si vede ora che i coefficienti di I valgono circa $10^{(-49)}$. Sembra proprio che le 3 radici siano reali.

Passiamo ora a un'equazione di quinto grado:

```
> solve(x^5+3*x^4-2*x^2+5*x-2=0,x);
```

$$\mathrm{RootOf}(\%1,\ index = 1),\ \mathrm{RootOf}(\%1,\ index = 2),\ \mathrm{RootOf}(\%1,\ index = 3),$$
$$\mathrm{RootOf}(\%1,\ index = 4),\ \mathrm{RootOf}(\%1,\ index = 5)$$
$$\%1 := \_Z^5 + 3\,\_Z^4 - 2\,\_Z^2 + 5\,\_Z - 2$$

Maple risponde che le soluzioni sono le sue radici. E' una tautologia. Pero', se uno e' capace, puo' riuscire a manipolare questi oggetti; e comunque li puo' valutare in forma decimale:

```
> evalf(%);
```

$$.4530021348,\ .4980252005 + .7820300577\,I,\ -2.224526268 + .4331058445\,I,$$
$$-2.224526268 - .4331058445\,I,\ .4980252005 - .7820300577\,I$$

si vede che una radice e' reale; le altre 4 no.

Passiamo a un'equazione non algebrica, detta anche trascendente. [Nota: un'equazione $\mathrm{P}(x) = 0$ dove P e' un polinomio, si dice algebrica; ogni altra equazione si dice trascendente]

```
> solve(exp(x)=5*x,x);
```

$$-\mathrm{LambertW}(\frac{-1}{5}),\ -\mathrm{LambertW}(-1,\ \frac{-1}{5})$$

Maple conosce le soluzioni in termini di certe funzioni "LambertW" che io non avevo mai sentito nominare.

Quanto valgono?

```
> evalf(%);
```

$$.2591711018,\ 2.542641358$$

ok.

Passiamo ora all'uso di "fsolve". Come detto "fsolve" usa metodi numerici e quindi non puo' risolvere equazioni contenenti simboli:

```
> fsolve(a*x^2+b*x+c=0,x);
```

```
Error, (in fsolve) {a, c, b} are in the equation, and are not solved
for
```

Possiamo pero' risolvere equazioni di 2o grado:

```
> fsolve(3*x^2-7*x+2 = 0, x);
```

$$.3333333333,\ 2.000000000$$

di terzo:

```
> fsolve(x^3-7*x+4,x);
```

$$-2.895106516,\ .6027049307,\ 2.292401585$$

in questo caso, vediamo senza problemi che vi sono 3 radici reali

di quinto:

```
> fsolve(x^5+3*x^4-2*x^2+5*x-2=0,x);
```

$$.4530021348$$

In questo caso, troviamo un'unica radice. "fsolve" trova solo soluzioni reali.

l'equazione trascendente di prima:

```
> fsolve(exp(x)=5*x,x);
```

$$.2591711018$$

"fsolve" ha trovato una sola soluzione. Questo perche' in genere "fsolve" usa un algoritmo che, da un dato punto di partenza, trova una soluzione. Unicamente nel caso di equazioni algebriche, "fsolve" contiene un metodo per trovare, dopo una radice, le altre eventualmente esistenti.

Se vogliamo trovarne una diversa, dobbiamo indicare a "fsolve" l'intervallo in cui andare a cercare la soluzione

In questo caso, scegliamo un intervallo che escluda la soluzione gia' trovata

```
> fsolve(exp(x)=5*x,x=1..10);
```

$$2.542641358$$

proviamo a dare un intervallo in cui ci sembra non vi siano radici:

```
> fsolve(exp(x)=5*x,x=3..5);
```

$$\mathrm{fsolve}(e^x = 5\,x,\ x,\ 3..5)$$

Maple risponde cosi'.

In genere, conviene usare "fsolve" dopo aver fatto un grafico delle funzioni che ci interessa uguagliare, in modo da avere un'idea su dove cercare le soluzioni:

```
> plot([exp(x),5*x],x=-1..6);
```